
Writing Your Own REXX
Functions — Simplified

I sn’t it interesting how IBM can provide
VSE with such great power but no

simple way of using it? A case in point is
REXX/VSE. With REXX/VSE you can use
a very powerful programming language to
create code in a short period of time. If you
read the REXX/VSE Reference Guide, you
will see that the first half of it actually
shows you how to use REXX/VSE and
provides you with many working examples.
So what’s in the last 200+ pages of the
manual? The last half of the manual pre-
sents an assortment of technical figures and
explanations. I suppose the intent was to
show the reader how to write a REXX
function, but in the end, the manual made
what should be a simple process into a very
complicated one. 

It would have been nice to actually be
shown an example of a working user-written
function, especially since there is no con-
sistency as to how the fields within the
various REXX control blocks are used. For
example, on page 280 of the REXX/VSE
Reference Guide we see “parameter 4”
described as being 32 bytes long, when in
reality it is a 4-byte pointer to the 32-byte
field. It also says that parameter 6 contains
a pointer, when in reality it contains a
pointer to another pointer. And on page 282
of that same manual, we see that length
fields really do contain length values. It
was the inconsistency and the mislabeling
of fields that resulted in lengthening the
actual development time of a simple
REXX/VSE function. It was only through
a lot of trial and error, and some help from
IBM, that I was able to successfully write a
REXX/VSE function. What I will show you
now should eliminate that same frustration
for you.

Now why might you want to write your
own functions? Well, perhaps you want to

augment the power of REXX/VSE by giving
it a facility that it doesn’t already have, such
as the ability to process VSAM KSDS files,
for example. There is no limit as to what
you can do, except for the limits within the
VSE environment itself.

In reality, there are only a few things that
you will want every function to do, including
the following:

1. Get the setting of a REXX variable.
2. Set a REXX variable to a new setting.
3. Set a return code.
4. Output to SYSLST or SYSLOG,

depending on the “ASSGN” setting.

That’s basically it. Now wouldn’t it have
been nice if IBM provided some simple
macro or application programming interface
(API) to help us out? Instead of defining all
of these control blocks and figuring out
values, it would be helpful to have all of
this work done for us. Also, because of the
confusing way that the manual shows how
to handle REXX control blocks, an API or
macro would eliminate the need to keep
referring to the manual every time a simple
function is necessary.

I did just that with the creation of a
macro called “REXXFUNC,” which allows
you to perform the following:

● BEGIN: Begin the function and allocate
and define REXX resources.

● END: Terminate the function and free
up allocated resources

● GETVAL: Get the value of a set variable
● SETVAL: Set the value of a REXX

variable
● GETPARM: Interrogate the passed

parameters by position (first, second, etc.)

// JOB CATAL
// EXEC LIBR, PARM=’A S=PRD2.TEST’
CATALOG TEST2.PROC REP=Y EOD=/+
input=’This is a test’
x=rexparmu(‘INPUT’,’OUTPUT’)
say ‘Output is: ‘ output
EXIT x
/+
/*
// LIBDEF *,SEARCH=PRD2.TEST
// EXEC REXX=TEST2
/&

FIGURE 1: JCL FOR RUNNING A TEST OF THE “REXPARMU” FUNCTION

● SAY: Output a message to SYSLOG,
SYSLST, or not at all.

This month, I will demonstrate the use of
a simple function (“rexparmu”) that will
read the passed parameters, set an output
parameter, output some messages, and ter-
minate with a return code. Figure 1 contains
the JCL for running a test of this function.

As you can see, the input field is set to
“This is a test”. The “rexparmu” function
will determine the first parameter passed,
get the value, and display it on SYSLOG.
It will also get the name of the second
parameter and set that to a specific string
of characters. It does this by using the
GETPARM function of REXXFUNC.

If you look at Figure 2, you will see the
source code that does all of the work. In
order to make re-entrant code, I have the
REXXFUNC macro allocate a block of
GETVIS, and this area is reused over and
over again. Because of this, a work regis-
ter is indicated in “AREA=” during the
initialization process.

I could have used a “WTO” macro, but
the advantage of using “SAY” is that it will
output to SYSLOG, SYSLST, or not at all,
depending on the “ASSGN” setting of the
invoking REXX program. Because of that,
this functionality has been provided.

As you can see, it is easy to produce and
debug a simple program. It is no longer
necessary to go through all the gyrations
described in the IBM manual; thus, a pro-
grammer can now focus on programming
instead of IBM control blocks.

The REXXFUNC macro can be down-
loaded from the NaSPA web site
(www.naspa.com) as filename VSE0201.txt.
To access, click on “Technical Support” and
then “Coding Examples.”

Happy coding!  

NaSPA member Leo J. Langevin is a systems
programmer with Connectivity Systems and
his career as a VSE developer spans two
millenniums. You can contact Leo via email
at leo@tcpip4vse.com.

REXPARMU REXXFUNC BEGIN,AREA=R8                                                

MVC RCVALUE(5),=CL5’0’ Indicate input problem                  
MVI MSG1+4,C’1’      Indicate parm 1                        
MVC PARMNUM,=F’1’    Here too                              
BAL R7,CHEKPARM       And check it                           
STM R3,R4,P1LEN      Save the length/pointer                 
MVI MSG1+4,C’2’ Indicate parm 2                        
MVC PARMNUM,=F’2’   Here too                              
BAL R7,CHEKPARM       And check it                           
STM R3,R4,P2LEN     Save the length/pointer                 
MVI RCVALUE,C’4’       Indicate input problem                  
ICM R1,15,P1LEN      Any input ?                            
BNP DOOUT              No...process the output                 
REXXFUNC GETVAL,PARMA=P1ADDR,PARMLEN=P1LEN,                   X        

BUFFER=BUFFER,BUFFLEN==AL4(L’BUFFER)                     
LTR R15,R15        Did it work ?                          
BZ SAYINPUT       Yes...proceed                          
REXXFUNC SAY,BUFFER=ERR1,BUFFLEN==AL4(L’ERR1)                  
B DOOUT           And proceed                            

SAYINPUT MVC MSG2D(80),BUFFER     Copy the buffer                        
LA  R14,16(R14)       Include the header                     
ST R14,LEN          Save the length                        
REXXFUNC SAY,BUFFER=MSG2,BUFFLEN=LEN                           

DOOUT ICM R1,15,P2LEN      Any output ?                           
BP  DOOUT2          Yes...proceed                          
LA R1,=CL6’OUTPUT’    Point here                             
ST R1,P2ADDR         Save this                              
MVC P2LEN(4),=F’6’     Update the length                      

DOOUT2 REXXFUNC SETVAL,PARMA=P2ADDR,PARMLEN=P2LEN,                   X        
BUFFER==C’The value has been set!’,BUFFLEN==F’23’                

MVI RCVALUE,C’8’       Indicate output problem                 
LTR R15,R15          Did it work ?                          
BZ  SAYOUT            Yes...proceed                          
REXXFUNC SAY,BUFFER=ERR2,BUFFLEN==AL4(L’ERR2)                  
B  REXXSET           And proceed                            

SAYOUT REXXFUNC SAY,BUFFER=MSG3,BUFFLEN==AL4(L’MSG3)                  
MVI RCVALUE,C’0’       No problem now                         

REXXSET DS 0H                                                            
REXXFUNC SETCODE,RC=RCVALUE,RCLEN==F’1’                        

REXPOUT REXXFUNC END                                                   
CHEKPARM DS    0H                                                            

REXXFUNC GETPARM,PARMNUM=PARMNUM                               
XR R3,R3           Init the length                        
LR R4,R1              Save the pointer                       
MVC MSG1D(7),=C’Not Set’ Default info                           
MVC LEN(4),=F’14’      Default length                         
LTR R15,R15             Did we find it ?                       
BNZ TELLPARM            No...proceed                           
LR  R3,R14             Regain the length                      
BCTR R14,0              Relative to 1                          
EX R14,COPY1          And copy the data                      
LA R14,8(R14)         Include header stuff                   
ST R14,LEN          Update the length                      

TELLPARM REXXFUNC SAY,BUFFER=MSG1,BUFFLEN=LEN                           
BR  R7                And proceed                            

COPY1 MVC MSG1D(0),0(R1)    Copy the data                          
LTORG ,              Literal pool                           

MSG1 DC C’PARM1: ‘                                                     
MSG1D  DS CL80                                                          
MSG2  DC C’Input contains: ‘                                             
MSG2D DS  CL80                                                          
ERR1   DC  C’Unable to read the input value’                                
ERR2   DC C’Unable to set the output value’                                
MSG3   DC C’Output parm contains a value’                                  
PARMNUM DS F            For parm searching                     
LEN   DS  F               For length passing                     
RCVALUE DS CL5               Return code goes in here                
P1LEN DS F                   Input parm info                        
P1ADDR DS A                                                             
P2LEN DS  F               Output parm info                       
P2ADDR DS  A                                                             
BUFFER DS  CL256             Generic I/O Buffer                     

END REXPARMU

FIGURE 2: SOURCE CODE FOR THE REXPARMU FUNCTION 


